

Министерство образования Республики Беларусь
Учреждение образования
«Белорусский государственный университет
информатики и радиоэлектроники»

Кафедра радиотехнических систем

В.Н. Левкович, О.В. Шабров, А.И. Бурак

***ВЫЧИСЛИТЕЛЬНЫЕ И КОММУНИКАЦИОННЫЕ
СРЕДСТВА РАДИОСИСТЕМ***

Лабораторный практикум
для студентов специальности
I-39 01 02 «Радиоэлектронные системы»
дневной формы обучения

В 2-х частях

Часть 2

Минск 2006

УДК 004.31(075.8)
ББК 32.973 я 73
Л 37

Рецензент:
доцент кафедры сетей и устройств телекоммуникаций БГУИР,
канд. техн. наук И.И. Астровский

Левкович В.Н.

Л 37 Вычислительные и коммуникационные средства радиосистем: Лаб. практикум для студ. спец. I-39 01 02 «Радиоэлектронные системы» дневн. формы обуч.: В 2 ч. Ч. 2 / В.Н. Левкович, О.В. Шабров, А.И. Бурак. – Мн: БГУИР, 2006. – 28 с.: ил.

ISBN 985-444-947-5 (ч.2)

Во вторую часть лабораторного практикума вошли теоретические сведения, примеры управляющих программ и варианты заданий для двух лабораторных работ по темам «Цифроаналоговое преобразование на основе широтно-импульсной модуляции» и «Аналого-цифровое преобразование».

Часть 1 вышла в БГУИР в 2004 г.

УДК 004.31(075.8)
ББК 32.973 я 73

ISBN 985-444-947-5 (ч.2)
ISBN 985-444-744-8

©Левкович В.Н., Шабров О.В.,
Бурак А.И., 2006
© БГУИР, 2006

СОДЕРЖАНИЕ

1. ЛАБОРАТОРНАЯ РАБОТА

Цифроаналоговое преобразование на основе широтно-импульсной модуляции (ШИМ)	3
1.1. Принцип цифроаналогового преобразования на основе ШИМ.....	3
1.2. Описание аппаратуры ШИМ	3
1.2.1. Общие сведения.....	3
1.2.2. Управляющий регистр CCRхCON	4
1.2.3. Структурная схема	5
1.2.4. Длительность импульса ШИМ	6
1.2.5. Последовательность настройки модуля CCR в ШИМ-режиме.....	7
1.3. Пример лабораторного задания и его решения	8
1.4. Требования к отчету	12
1.5. Контрольные вопросы.....	12
2. ЛАБОРАТОРНАЯ РАБОТА	
Аналого-цифровое преобразование	13
2.1. Структурная схема модуля аналого-цифрового преобразования.....	13
2.2. Управляющие регистры	14
2.2.1. Регистр ADCON0.....	14
2.2.2. Регистр ADCON1.....	15
2.3. Работа модуля АЦП	16
2.4. Временные требования к подключению канала АЦП.....	17
2.5. Выбор источника тактовых импульсов для АЦП.....	19
2.6. Настройка аналоговых входов.....	20
2.7. Аналого-цифровое преобразование	20
2.8. Выравнивание результата преобразования.....	21
2.9. Работа модуля АЦП в SLEEP-режиме микроконтроллера	22
2.10. Эффект сброса	23
2.11. Точность преобразования АЦП	23
2.12. Использование CCR-триггера.....	24
2.13. Передаточная функция модуля АЦП	24
2.14. Примеры процедур инициализации модуля АЦП.....	25
2.15. Пример задания на лабораторную работу.....	25
2.16. Требования к отчету.....	26
2.17. Контрольные вопросы.....	26
ЛИТЕРАТУРА.....	27

1. ЛАБОРАТОРНАЯ РАБОТА

Цифроаналоговое преобразование на основе широтно-импульсной модуляции (ШИМ)

Цель работы:

- 1) изучение принципов организации цифроаналогового преобразования на основе ШИМ;
- 2) приобретение практических навыков настройки и программирования встроенных аппаратных средств микроконтроллера для ШИМ.

1.1. Принцип цифроаналогового преобразования на основе ШИМ

На рис. 1 показан широтно-импульсно-модулированный сигнал. Он описывается тремя параметрами: амплитудой E , периодом T и длительностью t импульсов. При этом параметры E и T остаются постоянными, а t изменяется дискретно с точностью $1/2^N$, где N – разрядность кода для представления параметра t .

Принцип цифроаналогового преобразования (ЦАП) на основе ШИМ заключается в выделении с помощью фильтра нижних частот (ФНЧ) постоянной составляющей U_0 ШИМ-сигнала. При этом $U_0 = E t / T$.

Частота среза ФНЧ должна выбираться из условия требуемого подавления несущей частоты ($1/T$) ШИМ-сигнала. Если, к примеру, в качестве ФНЧ используется интегрирующая RC-цепочка с постоянной времени $\tau_{\text{ФНЧ}} = RC$, то для обеспечения пульсаций выходного сигнала меньше единицы младшего разряда ЦАП ($E/2^N$) постоянная времени ФНЧ должна выбираться из условия

$$RC \geq T \cdot 2^N.$$

При этом частота среза $F_{\text{ср}}$ ФНЧ может быть найдена из выражения

$$F_{\text{ср}} = 1 / 2\pi RC.$$




Рис. 1. Широтно-импульсно-модулированный сигнал

1.2. Описание аппаратуры ШИМ

1.2.1. Общие сведения

Широтно-импульсная модуляция в микроконтроллере PIC16F677 осуществляется с помощью встроенных аппаратных модулей CCP (захват/сравнение/ШИМ).

Каждый из двух модулей CCP содержит 16-разрядный регистр, который может использоваться в качестве:

- 16-разрядного регистра захвата данных;
- 16-разрядного регистра сравнения;
- двух 8-разрядных (ведущий и ведомый) регистров ШИМ.

Работа модулей CCP1 и CCP2 идентична, за исключением функционирования триггера специального события, поэтому в настоящем практикуме регистры модуля CCP будут обозначаться, как показано в табл.1.

Таблица 1

Обозначение	CCP1	CCP2	Примечание
CCPxCON	CCP1CON	CCP2CON	Управляющий регистр CCP
CCPRxH	CCPR1H	CCPR2H	Старший байт CCP
CCPRxL	CCPR1L	CCPR2L	Младший байт CCP
CCPx	CCP1	CCP2	Выход CCP

В табл. 2 указаны ресурсы, используемые модулем CCP, а в табл. 3 представлено взаимодействие между модулями CCP.

Таблица 2

Режим модуля CCP	Таймер
Захват	TMR1
Сравнение	TMR1
ШИМ	TMR2

Таблица 3

Режим CCP1	Режим CCP2	Взаимодействие
Захват	Захват	Базовый таймер TMR1
Захват	Сравнение	Модуль CCP, работающий в режиме сравнения, должен сбрасывать таймер TMR1 триггером специального события
Сравнение	Сравнение	Модули CCP, работающие в режиме сравнения, должны сбрасывать таймер TMR1 триггером специального события
ШИМ	ШИМ	Оба ШИМ имеют одинаковую частоту и фазу (базовый таймер TMR2)
ШИМ	Захват	Нет
ШИМ	Сравнение	Нет

1.2.2. Управляющий регистр CCPxCON

Регистры CCPxCON доступны по адресам 17h и 1Dh. Структура регистра отражена на рис. 2.

–	–	CCPxX	CCPxY	CCPxM3	CCPxM2	CCPxM1	CCPxM0
Бит 7	Бит 6	Бит 5	Бит 4	Бит 3	Бит 2	Бит 1	Бит 0

Рис. 2. Структура регистра CCPxCON

Все разряды регистра доступны для записи и чтения, по сигналу СБРОС они оказываются в состоянии логического нуля.

Биты 7 – 6 не используются, всегда читаются как '0'.

Биты 5 – 4 (ССРxX и ССРxY) являются двумя младшими битами 10-разрядного значения длительности импульса ШИМ (восемь старших битов DCxB9-DCxB2 находятся в ССРxL).

Биты 3 – 0 (ССРxM3:ССРxM0) определяют режим работы модуля ССРx.

Соответствие управляющих кодов ССРxM3:ССРxM0 и режимов работы ССР отражено в табл. 4.

Таблица 4

<i>ССРxM3: ССРxM0</i>	<i>Режим работы ССР</i>
0000	Модуль ССРx выключен (сброс модуля ССРx)
0001	Захват по каждому заднему фронту сигнала
0010	Захват по каждому переднему фронту сигнала
0011	Захват по каждому 4-му переднему фронту сигнала
0100	Захват по каждому 16-му переднему фронту сигнала
1000	Сравнение, устанавливает выходной сигнал (устанавливается флаг ССРIF в '1')
1001	Сравнение, сбрасывает выходной сигнал (устанавливается флаг ССРIF в '1')
1010	Сравнение, на выходной сигнал не влияет (устанавливается флаг ССРIF в '1')
1011	Сравнение, триггер специальных функций (устанавливается флаг ССРIF в T)
11xx	ШИМ-режим

В ШИМ-режиме модуля ССР вывод ССРx используется в качестве выхода 10-разрядного ШИМ. Так как вывод ССРx мультиплексирован с цифровым каналом порта ввода/вывода, то соответствующий бит направления TRIS должен быть сброшен в '0'.

Примечание. Очистка регистра ССРxCON вынудит перевести вывод ССРx в низкий логический уровень. Низкий логический уровень не является данными из защелки PORTC.

1.2.3. Структурная схема

На рис. 3 показана структурная схема модуля ССРx в ШИМ-режиме.

Период ШИМ определяется значением в регистре PR2 и может быть вычислен по формуле

$$\text{период ШИМ} = [(PR2)+1]*4*Tosc*(\text{коэффициент делителя TMR2}).$$

Когда значение TMR2 сравнивается с PR2, выполняются следующие действия:

- TMR2 сбрасывается в 00h;
- устанавливается высокий уровень сигнала на выводе ССРx (если скважность равна 0 %, то сигнал в высокий уровень устанавливаться не будет);
- модуль ШИМ начинает новый цикл, загружая значение из регистра ССРxL в ССРxH.

Примечание. Выходной делитель TMR2 не влияет на частоту ШИМ. Он может использоваться для отсчета времени, когда необходимо изменить скважность ШИМ.




Рис. 3. Структурная схема модуля CCPx в ШИМ-режиме

1.2.4. Длительность импульса ШИМ

Длительность импульса ШИМ определяется битами в регистрах CCPRxL и CCPxCON<5:4>. Для 10-разрядного ШИМ старшие восемь бит сохраняются в регистре CCPRxL, а младшие два бита – в регистре CCPxCON<5:4> (CCPRxL:CCPxCON<5:4>).

10-разрядное значение представляется битами DCxB9:DCxB0.

Для вычисления длительности сигнала высокого уровня можно воспользоваться следующей формулой:

длительность импульса ШИМ = (DCxB9:DCxB0) * T_{osc} * (коэффициент делителя TMR2).

Биты в регистре CCPRxL и CCPxCON<5:4> могут быть изменены в любое время, но значение в регистре CCPRxH не изменится, пока не произойдет соответствие PR2 и TMR2. В ШИМ-режиме регистр CCPRxH доступен только для чтения.

Регистр CCPxH и внутренняя 2-разрядная защелка образуют буфер ШИМ. Эффект буферизации необходим при записи нового значения длительности импульса ШИМ.

Когда значения CCPxH и 2-разрядной внутренней защелки соответствуют значению TMR2 и внутреннему 2-разрядному счетчику, в такте Q2 на выводе CCPx будет установлен низкий уровень сигнала (конец цикла ШИМ).

Максимальную разрядность N в битах ШИМ-сигнала с заданной частотой F_{PWM} для определенной частоты синхронизации F_{OSC} можно вычислить по формуле

$$N = \frac{\log\left(\frac{F_{OSC}}{F_{PWM}}\right)}{\log 2}.$$

Примечание. Если длительность импульса ШИМ больше периода ШИМ, вывод CCP1 не будет иметь низкий уровень сигнала, что позволяет реализовать скважность выходного сигнала 100 %.

Минимальная разрешающая способность каждого бита импульса ШИМ зависит от режима работы предделителя, как показано в табл. 5. Здесь T_{CY} – длительность командного цикла ($T_{CY}=4 T_{OSC}$).

Таблица 5

Коэффициент предделителя	T2CKPS1:T2CKPS0	Минимальное разрешение
1	00	T_{OSC}
4	01	T_{CY}
16	1x	$4T_{CY}$

1.2.5. Последовательность настройки модуля CCP в ШИМ-режиме

1. Установить период ШИМ в регистре PR2.
2. Установить длительность импульса в битах DCxB9:DCxB0;
3. Настроить вывод CCPx как выход, сбросив соответствующий бит TRIS.
4. Настроить предделитель и включить TMR2 в регистре T2CON.
5. Включить CCP в режиме ШИМ.

В табл. 6 представлено соответствие частоты и разрядности ШИМ при $F_{OSC} = 20\text{МГц}$. Здесь также показаны значения PR2 и коэффициента предделителя TMR2.

Таблица 6

Частота ШИМ, кГц	1,22	4,88	19,53	78,12	156,3	208,3
Коэффициент предделителя TMR2	16	4	1	1	1	1
Значение PR2	0xFF	0xFF	0xFF	0x3F	0x1F	0x17
Разрешение ШИМ (бит)	10	10	10	8	7	5.5

В SLEEP-режиме микроконтроллера TMR2 не будет инкрементироваться, и состояние модуля CCPx не изменяется. Если на выходе CCP присутствует определенный уровень сигнала, он будет удерживаться до тех пор, пока микроконтроллер находится в SLEEP-режиме. По выходу из SLEEP-режима модуль CCPx продолжит работать с текущего состояния.

После любого вида сброса модуль CCP выключен.

Ниже приведен пример процедуры инициализации модуля CCP1 в ШИМ-режиме.

```
BANK1      ;
MOVLW 0xFF ; таймер 2 загружаем значением 255 (это дает частоту
MOVWF PR2   ;ШИМ 19,53 кГц при тактовой частоте 20 МГц);
BANK0      ;
BCF CCP1CON,4 ; настройка длительности импульса ШИМ:
BCF CCP1CON,5 ; сбросить биты <5:4> регистра CCP1CON,
MOVLW 0x00   ; загрузка значения 0x00
MOVWF CCP1L  ; в регистр длительности импульса ШИМ
MOVLW B'00000100' ;
MOVWF T2CON  ;включение таймера 2 с предделителем =1:1
MOVLW B'00001111' ;
IORWF CCP1CON,F ;включение модуля CCP1 в режиме ШИМ
```

1.3. Пример лабораторного задания и его решения

Задание. Разработать алгоритм и составить программу формирования синусоидального сигнала частотой 100 Гц и амплитудой 2 В. Разрешение ШИМ 8 бит. Рассчитать требуемые номиналы ФНЧ. Аппаратные средства: демонстрационная плата DEMOGAMMA 2, внутрисхемный отладчик ICD 2.

Решение. Разобьем период формируемого сигнала на 36 дискретов и вычислим значения синуса в этих точках. Затем масштабируем полученные значения исходя из того, что максимальное значение формируемого сигнала, равное 5 В, при значении длительности импульса ШИМ будет равным FF, а в задании требуется 2 В. Следовательно, значение амплитуды 2 В при длительности импульса ШИМ будет формироваться равным $(FF \cdot 2) / 5 = 66$. Частота дискретизации при этом будет: $f_d = 100 \cdot 36 = 3600$ (Гц). Частоту ШИМ-сигнала желательно выбирать как можно более высокой для лучшей фильтрации нежелательных компонент. Максимально возможная частота ШИМ при тактовой частоте контроллера 20 МГц и разрешении ШИМ 8 бит равна 78,125 кГц. Желательно обеспечить целое количество периодов ШИМ сигнала на каждом отсчете формируемого сигнала. Поэтому возьмем соотношение f_d и $f_{\text{ШИМ}}$ равное 1:19. Тогда $f_{\text{ШИМ}} = 3600 \cdot 19 = 68400$ (Гц). Для обеспечения такой частоты ШИМ нужно запи-

сать в регистр PR2 значение 73. Точное значение частоты ШИМ при этом будет $f_{\text{ШИМ}} = (20\ 000\ 000/4)/73 = 68\ 493$ (Гц), что дает погрешность 0,13 %.

Ниже приведена программа, реализующая данную задачу.

```
LIST      p=16f877A      ;указание типа контроллера
#include  <p16f877A.inc>  ;подключение библиотечного файла с
                        ;описанием регистров микроконтроллера
__CONFIG _CP_OFF & _WDT_OFF & _BODEN_OFF & _PWRTE_ON & _HS_OSC &
_WRT_OFF & _LVP_OFF & _CPD_OFF ;описание конфигурации
```

```
BANK0 MACRO                ;создание макропоследовательностей
```

```
    BCF STATUS,RP1        ;
    BCF STATUS,RP0        ;ВЫБОР БАНКА 0
    EXITM
    ENDM
```

```
BANK1 MACRO
```

```
    BCF STATUS,RP1        ;
    BSF STATUS,RP0        ; ВЫБОР БАНКА 1
    EXITM
    ENDM
```

```
;**** ОПИСАНИЕ ПЕРЕМЕННЫХ ****
```

```
W_TEMP      EQU 0x71      ; переменные для временного хранения
STATUS_TEMP EQU 0x72      ; рабочих регистров при обработке
PCLATH_TEMP EQU 0x73      ; прерываний
    CBLOCK 0x20
```

```
COUNT_INT
```

```
COUNT_DELAY
```

```
SIN_TABLE_POINTER
```

```
FLAGS
```

```
SIN_TABLE
```

```
    ENDC
```

```
;*****НАЧАЛО ПРОГРАММЫ*****
```

```
    ORG 0x000            ;вектор сброса
    NOP                  ;требуется для ICD 2
    GOTO MAIN            ;переход на основную программу
    ORG 0x004            ; вектор прерывания
    MOVWF W_TEMP         ; сохранение регистра W
    MOVF STATUS,W        ;
    MOVWF STATUS_TEMP    ; сохранение регистра STATUS
    MOVF PCLATH,W        ;
    MOVWF PCLATH_TEMP    ; сохранение регистра PCLATH
```

```

BANK1 ;
BTFSS INTCON,T0IF ;флаг прерывания от таймера установлен ?
GOTO ENDINT ;если нет, то выход из прерывания
;если да, то обработать запрос на прерывание
BCF INTCON,T0IF ;обработчик прерывания от таймера
BANK0 ;
MOVLW 0xEC ;
MOVWF TMR0 ;перезагружаем таймер 0 числом 236
CALL READ_SIN_TABLE ;
MOVWF SIN_TABLE ;
CALL PWM_UPDATE ;
INCF SIN_TABLE_POINTER,F
MOVLW .36
SUBWF SIN_TABLE_POINTER,W
BTFSC STATUS,Z
CLRF SIN_TABLE_POINTER
ENDINT
MOVF PCLATH_TEMP,W ; восстановление регистров после
MOVWF PCLATH ; обработки прерывания
MOVF STATUS_TEMP,W ;
MOVWF STATUS ;
SWAPF W_TEMP,F ;
SWAPF W_TEMP,W ;
RETFIE ;
MAIN ;основная программа
CALL INITTIMER0 ;
CALL INIT_PWM ;
CLRF SIN_TABLE_POINTER
LOOP ;
GOTO LOOP ;

INITTIMER0
CLRF TMR0 ;обнуление таймера 0
MOVLW 0xEC ;начальное значение = 236
MOVWF TMR0 ;в таймер 0
BANK1 ;
MOVLW B'11000101' ;предделитель подключен к TMR0 (бит 3 = 0)
;тактовый сигнал синхронный (бит 5 = 0)
;внутренние подтягивающие резисторы выключены
;(бит 7 = 1)
MOVWF OPTION_REG ;записываем в OPTION_REG полученный байт

```

```

CLRF INTCON           ;очистка регистра прерываний
BSF INTCON,T0IE      ;разрешение прерывания от TMR0
BSF INTCON,PEIE     ;разрешение прерываний от устройств ввода-вывода
BSF INTCON,GIE      ;глобальное разрешение прерываний
BANK0                ;
RETURN              ;
INIT_PWM             ;
    BANK1            ;
    MOVLW 0x3F       ;
    MOVWF PR2        ;таймер 2 загружаем значением 63,
    BANK0            ;
    BCF CCP1CON,4    ;настройка длительности импульса ШИМ:
    BCF CCP1CON,5    ;сбросить биты <5:4> регистра CCP1CON,
    MOVLW 0x00       ;загрузка значения 0x00 (начальное значение)
    MOVWF CCP1L      ;в регистр длительности импульса ШИМ
    MOVLW B'00000100' ;
    MOVWF T2CON      ;включение таймера 2 с предделителем =1:1,
    MOVLW B'00001111' ;
    IORWF CCP1CON,F  ;включение модуля CCP1 в режиме ШИМ,
    RETURN           ;
READ_SIN_TABLE
    MOVF SIN_TABLE_POINTER,W
    CLRF PCLATH      ;
    ADDWF PCL,F      ;
    RETLW 0x00       ;код 1 отсчета
    RETLW 0x02       ;код 2 отсчета
    RETLW 0x08       ;код 3 отсчета
    RETLW 0x11       ;код 4 отсчета
    RETLW 0x1E       ;код 5 отсчета
    RETLW 0x2E       ;код 6 отсчета
    RETLW 0x40       ;код 7 отсчета
    RETLW 0x54       ;код 8 отсчета
    RETLW 0x69       ;код 9 отсчета
    RETLW 0x80       ;код 10 отсчета
    RETLW 0x96       ;код 11 отсчета
    RETLW 0xAB       ;код 12 отсчета
    RETLW 0xBF       ;код 13 отсчета
    RETLW 0xD1       ;код 14 отсчета
    RETLW 0xE1       ;код 15 отсчета
    RETLW 0xEE       ;код 16 отсчета
    RETLW 0xF7       ;код 17 отсчета

```

RETLW 0xFD	;код 18 отсчета
RETLW 0xFF	;код 19 отсчета
RETLW 0xFD	;код 20 отсчета
RETLW 0xF7	;код 21 отсчета
RETLW 0xEE	;код 22 отсчета
RETLW 0xE1	;код 23 отсчета
RETLW 0xD1	;код 24 отсчета
RETLW 0xBF	;код 25 отсчета
RETLW 0xAB	;код 26 отсчета
RETLW 0x96	;код 27 отсчета
RETLW 0x80	;код 28 отсчета
RETLW 0x69	;код 29 отсчета
RETLW 0x54	;код 30 отсчета
RETLW 0x40	;код 31 отсчета
RETLW 0x2E	;код 32 отсчета
RETLW 0x1E	;код 33 отсчета
RETLW 0x11	;код 34 отсчета
RETLW 0x08	;код 35 отсчета
RETLW 0x02	;код 36 отсчета

PWM_UPDATE ; смена параметров ШИМ-сигнала

MOVF SIN_TABLE,W	;
MOVWF CCP1L	;
RETURN	
END	; конец программы

1.4. Требования к отчету

1. Формулировка индивидуального задания, полученного у преподавателя.
2. Описание решения задачи с привлечением аналитических формул, временных диаграмм, схем функциональной и принципиальной, а также схем алгоритмов.
3. Текст отлаженной программы с комментариями.
4. Выводы по работе.

1.5. Контрольные вопросы

1. Поясните принцип ЦАП на основе ШИМ.
2. Поясните работу модуля ССР в режиме ШИМ по структурной схеме.
3. Приведите последовательность настройки модуля ССР в режиме ШИМ.
4. Как осуществляется формирование сигнала заданной формы с помощью ШИМ-сигнала?

2. ЛАБОРАТОРНАЯ РАБОТА

Аналого-цифровое преобразование

Цель работы:

- 1) изучение встроенного аппаратного модуля аналого-цифрового преобразования;
- 2) приобретение практических навыков настройки и программирования встроенных аппаратных средств микроконтроллера для аналого-цифрового преобразования.

2.1. Структурная схема модуля аналого-цифрового преобразования

Структурная схема модуля аналого-цифрового преобразования (АЦП) в PIC16F877 показана на рис. 4. Он преобразует входной аналоговый сигнал в соответствующий 10-разрядный цифровой код. Восемь аналоговых каналов мультиплексируются на одну схему выборки и хранения.




Рис. 4. Структурная схема модуля 10-разрядного АЦП

Входной аналоговый сигнал через коммутатор каналов заряжает внутренний конденсатор АЦП C_{HOLD} . Модуль АЦП преобразует напряжение, удерживаемое на конденсаторе C_{HOLD} , в соответствующий 10-разрядный цифровой код методом последовательного приближения. Источник верхнего и нижнего

опорного напряжения может быть программно выбран с выводов V_{DD} , V_{SS} , $AN3/V_{REF+}$ или $AN2/V_{REF-}$.

Допускается работа модуля АЦП в SLEEP-режиме микроконтроллера, при этом в качестве источника тактовых импульсов для АЦП должен быть выбран RC-генератор.

Для управления АЦП в микроконтроллере используется 4 регистра:

- регистр результата ADRESH (старший байт);
- регистр результата ADRESL (младший байт);
- регистр управления ADCON0;
- регистр управления ADCON1.

Регистр ADCON0 используется для настройки работы модуля АЦП, а с помощью регистра ADCON1 устанавливается, какие входы микроконтроллера будут использоваться модулем АЦП и в каком режиме (аналоговый вход или цифровой порт ввода/вывода).

2.2. Управляющие регистры

2.2.1. Регистр ADCON0

Регистр ADCON0 доступен по адресу 1Fh. Структура регистра отражена на рис. 5.

ADCS1	ADCS0	CHS2	CHS1	CHS0	GO/DONE	–	ADON
<i>Бит 7</i>	<i>Бит 6</i>	<i>Бит 5</i>	<i>Бит 4</i>	<i>Бит 3</i>	<i>Бит 2</i>	<i>Бит 1</i>	<i>Бит 0</i>

Рис. 5. Структура регистра ADCON0

Все разряды регистра доступны для записи и чтения, по сигналу СБРОС они оказываются в состоянии логического нуля.

Биты 7 – 6: ADCS1-ADCS0: выбор источника тактового сигнала:

- 00 = $F_{osc}/2$,
- 01 = $F_{osc}/8$,
- 10 = $F_{osc}/32$,
- 11 = F_{RC} (внутренний RC-генератор модуля АЦП).

Биты 5 – 3: CHS2-CHS0: выбор аналогового канала:

- 000 = канал 0, (AN0),
- 001 = канал 1, (AN1),
- 010 = канал 2, (AN2),
- 011 = канал 3, (AN3),
- 100 = канал 4, (AN4),
- 101 = канал 5, (AN5),
- 110 = канал 6, (AN6),
- 111 = канал 7, (AN7).

Бит 2: GO/-DONE: бит статуса модуля АЦП, если ADON = 1, модуль АЦП выполняет преобразование (установка бита вызывает начало преобразования); ADON = 0 – состояние ожидания (аппаратно сбрасывается по завершению преобразования)

Бит 1: не используется, читается как '0'.

Бит 0: ADON: бит включения модуля АЦП:

1 – модуль АЦП включен,

0 – модуль АЦП выключен и не потребляет ток.

2.2.2. Регистр ADCON1

Регистр ADCON1 доступен по адресу 9Fh. Структура регистра отражена на рис. 6.

ADFM	–	–	–	PCFG3	PCFG2	PCFG1	PCFG0
<i>Бит 7</i>	<i>Бит 6</i>	<i>Бит 5</i>	<i>Бит 4</i>	<i>Бит 3</i>	<i>Бит 2</i>	<i>Бит 1</i>	<i>Бит 0</i>

Рис. 6. Структура регистра ADCON1

Все разряды регистра доступны для записи и чтения, по сигналу СБРОС они оказываются в состоянии логического нуля.

Бит 7: ADFM: формат сохранения 10-разрядного результата (см. рис. 10):

1 – правое выравнивание, 6 старших бит ADRESH читаются как '0',

0 – левое выравнивание, 6 младших бит ADRESL читаются как '0'.

Биты 6 – 4: не используются, читаются как '0'.

Биты 3 – 0: PCFG3:PCFG0: управляющие биты настройки каналов АЦП, соответствие комбинаций значений разрядов и режимов работы каналов АЦП отражены в табл. 7.

Таблица 7

<i>PCFG3: PCFG0</i>	<i>AN7</i>	<i>AN6</i>	<i>AN5</i>	<i>AN4</i>	<i>AN3</i>	<i>AN2</i>	<i>AN1</i>	<i>AND</i>	<i>VR_{EF+}</i>	<i>VR_{EF-}</i>	<i>Кан.</i>
0000	A	A	A	A	A	A	A	A	AV _{DD}	AV _{SS}	8/0
0001	A	A	A	A	VREF+	A	A	A	AN3	AV _{SS}	7/1
0010	D	D	D	A	A	A	A	A	AV _{DD}	AV _{SS}	5/0
0011	D	D	D	A	VREF+	A	A	A	AN3	AV _{SS}	4/1
0100	D	D	D	D	A	D	A	A	AV _{DD}	AV _{SS}	3/0
0101	D	D	D	D	VREF+	D	A	A	AN3	AV _{SS}	2/1
011x	D	D	D	D	D	D	D	D	AV _{DD}	AV _{SS}	0/0
1000	A	A	A	A	VREF+	VREF-	A	A	AN3	AN2	6/2
1001	D	D	A	A	A	A	A	A	AV _{DD}	AV _{SS}	6/0
1010	D	D	A	A	VREF+	A	A	A	AN3	AV _{SS}	5/1
1011	D	D	A	A	VREF+	VREF-	A	A	AN3	AN2	4/2
1100	D	D	D	A	VREF+	VREF-	A	A	AN3	AN2	3/2
1101	D	D	D	D	VREF+	VREF-	A	A	AN3	AN2	2/2
1110	D	D	D	D	D	D	D	A	AV _{DD}	AV _{SS}	1/0
1111	D	D	D	D	VREF+	VREF-	D	A	AN3	AN2	1/2

В табл. 7 символом А обозначен аналоговый вход, символом D – цифровой канал ввода/вывода.

В последнем столбце табл. 7 указано число аналоговых каналов, доступных для выполнения преобразования, и число входов источника опорного напряжения.

При сбросе микроконтроллера все выводы, мультиплексированные с модулем АЦП (ANx), настраиваются как аналоговые входы.

2.3. Работа модуля АЦП

Когда преобразование завершено, 10-разрядный результат аналого-цифрового преобразования записывается в регистры ADRESH и ADRESL, после чего сбрасывается бит GO/DONE (ADCON0<2>) и устанавливается флаг прерывания ADIF.

После включения и настройки АЦП необходимо выбрать рабочий аналоговый канал. Соответствующие биты TRIS аналоговых каналов должны настраивать канал порта ввода/вывода на вход. Перед началом преобразования необходимо выдержать временную паузу, расчет которой приведен далее.

Рекомендуется следующая последовательность действий для работы с АЦП:

1. Настроить модуль АЦП:

- настроить выводы как аналоговые входы, входы V_{REF} или цифровые каналы ввода/вывода (ADCON1);
- выбрать входной канал АЦП (ADCON0);
- выбрать источник тактовых импульсов для АЦП (ADCON0);
- включить модуль АЦП (ADCON0).

2. Настроить прерывание от модуля АЦП (если необходимо):

- сбросить бит ADIF в '0';
- установить бит ADIE в '1';
- установить бит PEIE в '1';
- установить бит GIE в '1'.

3. Выдержать паузу, необходимую для зарядки конденсатора C_{HOLD} .

4. Начать аналого-цифровое преобразование:

- установить GO/DONE бит в '1' (ADCON0).

5. Ожидать окончания преобразования:

- ждать, пока бит GO/DONE не будет сброшен в '0'; ИЛИ;
- ожидать прерывание по окончании преобразования.

6. Считать результат преобразования из регистров ADRESH, ADRESL, сбросить бит ADIF в '0', если это необходимо.

7. Для следующего преобразования необходимо выполнить шаги, начиная с п. 1 или 2. Время преобразования одного бита определяется как

время T_{AD} . Минимальное время ожидания перед следующим преобразованием должно составлять не менее $2T_{AD}$.

На рис. 7 показана последовательность преобразования аналогового сигнала. Время заряда C_{HOLD} – интервал времени, в течение которого на внутренний конденсатор АЦП подается внешний сигнал. Время преобразования равно $12 T_{AD}$, отсчет начинается с момента установки в '1' бита GO.




Рис. 7. Последовательность преобразования аналогового сигнала

Сумма этих двух временных интервалов является длительностью полного цикла преобразования.

Существует минимальный интервал времени, в течение которого внешний сигнал подается на внутренний конденсатор C_{HOLD} , чтобы гарантировать требуемую точность АЦП.

2.4. Временные требования к подключению канала АЦП

Для обеспечения необходимой точности преобразования конденсатор C_{HOLD} должен успевать полностью заряжаться до уровня входного напряжения.

Схема аналогового входа АЦП показана на рис. 8. Сопротивления R_S и R_{SS} непосредственно влияют на время зарядки конденсатора C_{HOLD} . Величина сопротивления ключа выборки R_{SS} зависит от напряжения питания V_{DD} (рис. 8). Максимальное рекомендуемое значение внутреннего сопротивления источника аналогового сигнала 10 кОм. При меньших значениях сопротивления источника сигнала меньше и суммарное время преобразования.

После того как будет выбран один из нескольких аналоговых входных каналов, но прежде, чем будет производиться преобразование, должно пройти определенное время. Для нахождения данного времени необходимо воспользоваться уравнением (1). Это уравнение дает результат с ошибкой в $1/2 \text{ LSb}$. Ошибка в $1/2 \text{ LSb}$ – это максимальная погрешность, позволяющая функционировать модулю АЦП с необходимой точностью.

Временная задержка T_{ACQ} определяется суммарным временем задержки усилителя, заряда конденсатора C_{HOLD} и температурным коэффициентом:

$$T_{ACQ} = T_{AMP} + T_C + T_{COF} . \quad (1)$$

Минимальное время заряда конденсатора C_{HOLD} :

$$T_C = -120 \text{ пФ} (1 \text{ кОм} + R_S + R_{SS}) \text{ Ln} (1/2047). \quad (2)$$

В следующем примере показано вычисление минимального значения времени T_{ACQ} . Вычисления основываются на следующих исходных данных:

$$C_{HOLD} = 120 \text{ пФ},$$

$$R_S = 10 \text{ кОм},$$

ошибка преобразования $< 1/2 \text{ Lsb}$,

$$V_{DD} = 5\text{В} \Rightarrow R_{SS} = 7 \text{ кОм} \text{ (см. график на рис. 8)},$$

температура = 50°C (максимально возможная),

$$V_{HOLD} = 0 \text{ В} \text{ и } t = 0.$$

Вычисление минимального значения времени T_{ACQ} (случай 1):

$$T_{ACQ} = T_{AMP} + T_C + T_{COFF} .$$

Температурный коэффициент необходимо использовать только при рабочей температуре более 25°C .

$$T_{ACQ} = 2 \text{ мкс} + T_C + [(Температура - 25^\circ\text{C})(0,05 \text{ мкс}/^\circ\text{C})].$$

$$T_C = -C_{HOLD} (R_{IC} + R_{SS} + R_S) \text{ Ln}(1/2047) = \\ = -120 \text{ пФ} (1 \text{ кОм} + 7 \text{ кОм} + 10 \text{ кОм}) \text{ Ln}(0,0004885) = 16,47 \text{ мкс}.$$

$$T_{ACQ} = 2 \text{ мкс} + 16,47 \text{ мкс} + [(50^\circ\text{C} - 25^\circ\text{C})(0,05 \text{ мкс}/^\circ\text{C})] = 19,72 \text{ мкс}.$$

Для более точного представления влияния внутреннего сопротивления источника сигнала на время заряда конденсатора C_{HOLD} рассмотрим случай 2, отличающийся от первого тем, что внутреннее сопротивление источника сигнала равно 50 Ом .

Вычисление минимального значения времени T_{ACQ} (случай 2):

$$T_{ACQ} = T_{AMP} + T_C + T_{COFF} ;$$


$$T_{ACQ} = 2 \text{ мкс} + T_C + [(Температура - 25^\circ\text{C})(0,05 \text{ мкс}/^\circ\text{C})] ;$$

$$T_C = -C_{HOLD} (R_{IC} + R_{SS} + R_S) \text{ Ln}(1/2047) = \\ = -120 \text{ пФ} (1 \text{ кОм} + 7 \text{ кОм} + 50 \text{ Ом}) \text{ Ln}(0,0004885) = 7,36 \text{ мкс}.$$

$$T_{ACQ} = 2 \text{ мкс} + 7,36 \text{ мкс} + [(50^\circ\text{C} - 25^\circ\text{C})(0,05 \text{ мкс}/^\circ\text{C})] = 10,61 \text{ мкс}.$$

Примечания:

1. Напряжение источника опорного напряжения V_{REF} не влияет на уравнение.
2. Конденсатор C_{HOLD} после преобразования не разряжается.
3. Максимальное рекомендуемое значение внутреннего сопротивления источника аналогового сигнала 10 кОм . Это необходимо для компенсации внутреннего тока утечки.
4. После того как преобразование завершено, необходимо программно обеспечить задержку не менее $2T_{AD}$, прежде чем начать следующее преобразование. В течение этого времени конденсатор C_{HOLD} не подключен к выбранному входному каналу АЦП.



C_{PIN} – входная емкость; V_T – пороговое напряжение; $I_{LEAKAGE}$ – ток утечки вывода;
 R_{IC} – сопротивление соединения; SS – переключатель защелки; C_{HOLD} – конденсатор защелки.

Рис. 8. Схема аналогового входа АЦП

2.5. Выбор источника тактовых импульсов для АЦП

Время получения одного бита результата равно T_{AD} . Для 10-разрядного результата требуется как минимум 11,5 T_{AD} . Параметры тактового сигнала для АЦП определяются программно, T_{AD} может принимать следующие значения: $2T_{OSC}$, $8T_{OSC}$, $32T_{OSC}$, 2 – 6 мкс от внутреннего RC-генератора модуля АЦП.

Для получения корректного результата преобразования необходимо выбрать источник тактового сигнала АЦП, обеспечивающий время T_{AD} не менее 1,6 мкс.

В табл. 8 указано максимальное значение тактовой частоты микроконтроллера для каждого режима синхронизирующего сигнала АЦП.

Таблица 8

Источник импульсов АЦП		Рабочая частота F_{OSC}			
Источник	ADCS1: ADCS0	20 МГц	5 МГц	1,25 МГц	333,33 кГц
$2T_{OSC}$	00	100 нс ⁽²⁾	400 нс ⁽²⁾	1,6 мкс	6 мкс
$8T_{OSC}$	01	400 нс ⁽²⁾	1,6 мкс	6,4 мкс	24 мкс ⁽³⁾
$32T_{OSC}$	10	1,6 мкс	6,4 мкс	25,6 мкс ⁽³⁾	96 мкс ⁽³⁾
RC-генератор	11	2 – 6 мкс ^(1,4)	2 – 6 мкс ^(1,4)	2 – 6 мкс ^(1,4)	2 – 6 мкс ⁽¹⁾

Примечания:

1. Типовое значение времени T_{AD} RC-генератора АЦП равно 4 мкс.
2. Это значение выходит за пределы минимально допустимого времени T_{AD} .
3. Для более точного преобразования рекомендуется выбрать другой источник тактовых импульсов.
4. Когда тактовая частота микроконтроллера больше 1 МГц, рекомендуется использовать RC-генератор АЦП только для работы в SLEEP-режиме.

2.6. Настройка аналоговых входов

Регистры ADCON1, TRISA и TRISE отвечают за настройку выводов АЦП. Если выходы микросхемы настраиваются как аналоговые входы, то при этом должны быть установлены соответствующие биты в регистре TRIS. Если соответствующий бит сброшен в '0', вывод микросхемы настроен как цифровой выход. Модуль АЦП функционирует независимо от состояния бит CHS2:CHS0 и бит TRIS.

При чтении содержимого регистра порта нули будут установлены в тех разрядах, которые были настроены как аналоговые входы. Настроенные на цифровой вход каналы будут преобразовывать входные аналоговые уровни в цифровые, что, однако, не окажет влияния на точность преобразования.

Значения напряжений, подаваемых на выходы, настроенные как аналоговые входы, включая выходы AN7:AN0, могут влиять на ток потребления входного буфера, который может выйти за пределы значений, оговоренных в технической спецификации.

2.7. Аналого-цифровое преобразование

В примере ниже показана последовательность действий для работы с АЦП. Выводы настроены как аналоговые входы. Источник опорного напряжения – AV_{DD} , AV_{SS} . Разрешены прерывания от модуля АЦП. Источником импульсов преобразования является RC-генератор АЦП. Аналого-цифровое преобразование выполняется с вывода AN0.

Примечание. Бит GO/DONE и бит включения АЦП должны устанавливаться разными командами.

Сброс бита GO/DONE в '0' во время преобразования приведет к его прекращению. При этом регистры результата (ADRESH:ADRESL) не изменят своего содержимого. После досрочного завершения преобразования необходимо обеспечить временную задержку $2T_{AD}$. Выдержав требуемую паузу, можно начать новое преобразование установкой бита GO/DONE в '1'.

На рис. 9 показана последовательность получения результата после установки бита GO/DONE в '1'.

Пример выполнения преобразования АЦП.

BSF	STATUS, RPO	; Выбрать банк 1
CLRF	ADCON1	; Настроить входы АЦП
BSF	PIE1, ADIE	; Разрешить прерывания от АЦП
BCF	STATUS, RPO	; Выбрать банк 0
MOVLW	0xC1	; Тактовые импульсы от RC-генератора АЦП,
MOVWF	ADCON0	; Включить АЦП, выбрать канал 0
BCF	PIR1, ADIF	; Сбросить флаг прерываний от АЦП
BSF	INTCON, PEIE	; Разрешить периферийные прерывания
BSF	INTCON, GIE	; Разрешить прерывания в системе

- ; Выдержать паузу, необходимую для заряда внутреннего конденсатора C_{HOLD} , затем начинать преобразование АЦП.
- BSF ADCON0, GO ; Старт преобразования
- ; Ожидать установку флага ADIF или сброс бита GO/-DONE по завершению преобразования




Рис. 9. Последовательность получения результата АЦП

2.8. Выравнивание результата преобразования

10-разрядный результат преобразования сохраняется в спаренном 16-разрядном регистре ADRESH:ADRESL. Запись результата преобразования может выполняться с правым или левым выравниванием, в зависимости от значения бита ADFM (рис. 10). Не задействованные биты регистра ADRESH:ADRESL читаются как '0'. Если модуль АЦП выключен, то 8-разрядные регистры ADRESH и ADRESL могут использоваться как регистры общего назначения.




Рис. 10. Выравнивание результата аналого-цифрового преобразования

2.9. Работа модуля АЦП в SLEEP-режиме микроконтроллера

Модуль АЦП может работать в SLEEP-режиме микроконтроллера при условии, что источником импульсов преобразования АЦП будет внутренний RC-генератор (ADCS1:ADCS0=11). При выборе RC-генератора импульсов модуль АЦП делает задержку в один машинный цикл перед началом преобразования. Это позволяет программе пользователя выполнить команду SLEEP, тем самым уменьшить «цифровой шум» во время преобразования. После завершения преобразования аппаратно сбрасывается бит GO/DONE в '0', результат преобразования сохраняется в регистре ADRES. Если разрешено прерывание от АЦП, то микроконтроллер выйдет из режима SLEEP. Если же прерывание было запрещено, то после преобразования модуль АЦП будет выключен, хотя бит ADON останется установленным.

Если был выбран другой источник тактовых импульсов АЦП (не внутренний RC-генератор), то выполнение программой инструкции SLEEP прервет процесс преобразования и выключит модуль АЦП, оставив установленным бит ADON. Выключение модуля АЦП уменьшит ток потребления микроконтроллера.

Примечание. Для работы модуля АЦП в SLEEP-режиме необходимо выбрать внутренний RC-генератор (ADCS1:ADCS0=11), инструкция SLEEP должна быть выполнена сразу после команды, устанавливающей бит GO/DONE в '1'.

2.10. Эффект сброса

При сбросе микроконтроллера значения всех его регистров устанавливаются по умолчанию. Сброс выключает модуль АЦП, а также останавливает процесс преобразования, если он был начат. Все выводы, используемые модулем АЦП, настраиваются как аналоговые входы. Регистры ADRESH:ADRESL после сброса POR будет содержать неизвестное значение, а после остальных видов сброса не изменят своего значения.

2.11. Точность преобразования АЦП

Абсолютная точность АЦП определяется суммарной ошибкой, исходя из ошибки дискретизации, интегральной ошибки, ошибки шкалы, ошибки смещения и монотонности. Суммарная ошибка определяется как максимальный разброс между текущим и идеальным результатом для любого значения. Абсолютная ошибка АЦП меньше ± 1 значащего бита при $V_{DD} = V_{REF}$, но она возрастает при отклонении V_{REF} от V_{DD} .

В некотором диапазоне напряжений на аналоговом входе цифровой результат будет один и тот же. Это возникает из-за дискретизации, которая неизбежна при преобразовании аналоговой величины в цифровую форму. Ошибка дискретизации составляет $\pm 1/2$ значащего бита, и единственный способ уменьшить ее – увеличить разрядность АЦП.

Ошибку смещения составляет разность между результатом первого преобразования и идеальным значением. Эта ошибка сдвигает всю передаточную функцию и может быть учтена при помощи калибровки. Ошибка вносится в результате наложения токов утечки и выходного сопротивления источника сигнала.

Ошибка усиления измеряется как максимальное отклонение результата, скорректированного с учетом ошибки смещения. Эта ошибка проявляется в виде изменения наклона передаточной функции. Ошибка усиления может быть откалибрована и учтена.

Ошибка линейности определяется как разница в приращении входного напряжения для получения одинакового приращения выходного кода и не поддается калибровке. Интегральная ошибка вычисляется как отклонение результата, скорректированного с учетом ошибки усиления.

В системах с низкой тактовой частотой предпочтительно использование встроенного RC-генератора. В системах с высокой рабочей частотой следует использовать тактовый сигнал от основного генератора. Предпочтительно использовать АЦП с T_{AD} не больше 8 мкс, но не меньше рекомендованного нижнего предела. Использование тактового сигнала от основного генератора позволяет снизить влияние шумов от переключения внутренних вентилях, т.к. переключение логики АЦП происходит синхронно с другими устройствами, что невозможно при использовании встроенного RC-генератора. Если каналы цифрового ввода-вывода постоянно активны, потеря точности из-за шумов при переключении может быть значительной.

В случае использования АЦП в SLEEP-режиме источником тактового сигнала должен быть встроенный RC-генератор. В этом режиме отсутствуют цифровые шумы, т.к. другие узлы микроконтроллера остановлены, поэтому точность преобразования получается высокой.

2.12. Использование SSP-триггера

Аналого-цифровое преобразование может быть запущено при помощи «триггера специального события» модуля SSP. Для этого необходимо, чтобы биты SSPxM3:SSPxM0 (SSPxCOM<3:0>) были запрограммированы как 1011 и был включен модуль АЦП (бит ADON должен быть установлен в '1'). При срабатывании триггера бит GO/DONE будет установлен в '1', тем самым запуская преобразование, а содержимое таймера TMR1 будет обнулено. Таймер сбрасывается и автоматически повторяет запуск преобразования через определенные промежутки времени. Пользователю необходимо будет только вовремя считывать содержимое регистра ADRES. До начала преобразования необходимо выбрать соответствующий аналоговый канал, прежде чем «триггер специального события» вызовет установку бита GO/DONE.

При выключенном модуле АЦП (бит ADON сброшен в '0') сигнал «триггера специального события» игнорируется, но таймер TMR1 продолжает работать и сбрасываться.

Если значение входного напряжения превышает на 0,3 В величину порога питающих напряжений (V_{SS} и V_{DD}), то точность преобразования выйдет за пределы значений, оговоренных в спецификации.

Иногда для сглаживания пульсаций входного сигнала на вход АЦП добавляется внешняя RC-цепочка. Значение сопротивления R должно выбираться так, чтобы общее сопротивление источника сигнала было в пределах рекомендованной величины 10 кОм. Любой внешний электронный компонент, подключенный к аналоговому входу (конденсатор, стабилитрон и др.), должен иметь низкий ток утечки через вывод.

2.13. Передаточная функция модуля АЦП

Идеальная функция модуля АЦП работает по следующему правилу: первый бит значения измеряемой аналоговой величины будет установлен, если входное напряжение (V_{AIN}) на аналоговом входе будет равно 1 Lsb ($V_{REF}/1024$) (рис. 11).




Рис. 11. Идеальная передаточная функция модуля АЦП

2.14. Примеры процедур инициализации модуля АЦП

Инициализация модуля АЦП без использования прерывания от АЦП.

```
BSF      STATUS, RP0      ;выбрать банк 1
MOVLW   B'00001110'      ;
MOVWF   ADCON1           ;настроить входы АЦП
BCF     STATUS, RP0      ;выбрать банк 0
MOVLW   B'00000001'      ;тактовая частота АЦП  $F_{osc}/2$ , выбран канал 0,
MOVWF   ADCON0           ; АЦП включен;
```

Инициализация модуля АЦП с использованием прерываний от АЦП.

BSF	STATUS, RP0	;выбран банк 1,
MOVLW	B'00001110'	;
MOVWF	ADCON1	;настроить входы АЦП,
BSF	PIE1, ADIE	;разрешить прерывания от АЦП,
BCF	STATUS, RP0	;выбрать банк 0,
MOVLW	0xC1	;тактовые импульсы от RC-генератора АЦП,
MOVWF	ADCON0	; АЦП включен, выбран канал 0,
BCF	PIR1, ADIF	;сбросить флаг прерываний от АЦП,
BCF	INTCON, PEIE	;разрешены периферийные прерывания,
BCF	INTCON, GIE	;разрешены прерывания в системе вообще.

2.15. Пример задания на лабораторную работу

Разработать алгоритм и составить программу, выполняющую аналого-цифровое преобразование сигнала на входе RA0 микроконтроллера PIC16F877A.

Результат преобразования вывести на двухразрядный семисегментный индикатор.

Частота дискретизации $f_d = 10$ Гц, период обновления информации на индикаторе 1 с.

Источник опорного напряжения – напряжение питания микроконтроллера.

Аппаратные средства: демонстрационная плата DEMOGAMMA 2, внутрисхемный отладчик ICD 2.

2.16. Требования к отчету

1. Формулировка индивидуального задания, полученного у преподавателя.
2. Описание решения задачи с привлечением аналитических формул, временных диаграмм, схем функциональной и принципиальной, а также схем алгоритмов.
3. Текст отлаженной программы с комментариями.
4. Выводы по работе.

2.17. Контрольные вопросы

1. Поясните принцип работы модуля АЦП по структурной схеме.
2. Прокомментируйте режимы работы модуля АЦП, устанавливаемые с помощью управляющих регистров.
3. Приведите последовательность инициализации модуля АЦП.
4. Дайте временные характеристики процесса аналого-цифрового преобразования с помощью встроенного модуля АЦП
5. Назовите и поясните факторы, влияющие на точность аналого-цифрового преобразования.

ЛИТЕРАТУРА

1. Левкович В.Н. Вычислительные и коммуникационные средства радиосистем: Лаб. практикум для студ. спец. I-39 01 02 «Радиоэлектронные системы»: В 2 ч. Ч. 1 /В.Н. Левкович, А.И. Бурак. – Мн: БГУИР, 2004. – 59 с.

2. PIC16F87X. Однокристальные 8-разрядные FLASH CMOS микроконтроллеры компании Microchip technology incorporated: Пер. с англ. – М.: ООО «Микрочип», 2002. – 184 с. www.microchip.ru

3. Бурак А.И., Левкович В.Н. Интегрированная среда MPLab IDE разработки программ для микроконтроллеров PICmicro фирмы Microchip: Метод. пособие к лабораторным работам по курсу «Цифровые и микропроцессорные устройства». – Мн: БГУИР, 2003. – 31 с.

4. Левкович В.Н. и др. Конструирование программ на Ассемблере для микроконтроллеров семейства PICmicro: Учеб. пособие по курсу «Цифровые и микропроцессорные устройства» для студ. спец. I-39 01 01 «Радиотехника» и I-39 01 02 «Радиоэлектронные системы» всех форм обуч. – Мн: БГУИР, 2004. – 80 с.

Библиотека БГУИР

Учебное издание

Левкович Василий Николаевич,
Шабров Олег Васильевич,
Бурак Андрей Иосифович

**ВЫЧИСЛИТЕЛЬНЫЕ И КОММУНИКАЦИОННЫЕ
СРЕДСТВА РАДИОСИСТЕМ**

Лабораторный практикум
для студентов специальности
I-39 01 02 «Радиоэлектронные системы»
дневной формы обучения

В 2-х частях
Часть 2

Редактор Т.Н. Крюкова
Корректор Н.В. Гриневич

Подписано в печать
Гарнитура «Таймс».
Уч.-изд. л. 1,7.

Формат 60x84 1/16.
Печать ризографическая.
Тираж 125 экз.

Бумага офсетная.
Усл. печ. л. 1,74.
Заказ 15.

Издатель и полиграфическое исполнение: Учреждение образования
«Белорусский государственный университет информатики и радиоэлектроники»
ЛИ №02330/0056964 от 01.04.2004. ЛП №02330/0131518 от 30.04.2004.
220013, Минск, П. Бровки, 6